

Programming Languages: Functional Programming

1. Introduction to Haskell: Value, Functions, And Types

Shin-Cheng Mu

Autumn 2025

A Quick Introduction to Haskell

- We will mostly learn some syntactical issues, but there are some important messages too.
- Most of the materials today are adapted from the book *Introduction to Functional Programming using Haskell* by Richard Bird. Prentice Hall 1998.
- References to more Haskell materials are on the course homepage.

Course Materials and Tools

- Course homepage: <https://cool.ntu.edu.tw/courses/51303>
 - Announcements, slides, assignments, additional materials, etc.
- We will be using the Glasgow Haskell Compiler (GHC).
 - A Haskell compiler written in Haskell, with an interpreter that both interprets and runs compiled code.
 - See the course homepage for instructions for installation and other info.

Function Definition

- A function definition consists of a type declaration, and the definition of its body:

```
square    :: Int → Int
square x  = x × x
```

```
smaller   :: Int → Int → Int
smaller x y = if x ≤ y then x else y
```

- The GHCi interpreter evaluates expressions in the loaded context:

```
? square 3768
14197824
? square (smaller 5 (3+4))
25
```

1 Values and Evaluation

Evaluation

One possible sequence of evaluating (simplifying, or reducing) *square* (3 + 4):

```
square (3 + 4)
= { definition of + }
square 7
= { definition of square }
7 × 7
= { definition of × }
49
```

Another Evaluation Sequence

- Another possible reduction sequence:

```
square (3 + 4)
= { definition of square }
(3 + 4) × (3 + 4)
= { definition of + }
7 × (3 + 4)
= { definition of + }
7 × 7
= { definition of × }
49
```

- In this sequence the rule for *square* is applied first. The final result stays the same.
- Do different evaluations orders always yield the same thing?

A Non-terminating Reduction

- Consider the following program:

```
three  :: Int → Int
three x = 3
infinity :: Int
infinity = infinity + 1
```

- Try evaluating *three infinity*. If we simplify *infinity* first:

```
three infinity
= { definition of infinity }
three (infinity + 1)
= three ((infinity + 1) + 1)...
```

- If we start with simplifying *three*:

```
three infinity
= { definition of three }
3
```

Evaluation Order

- There can be many other evaluation orders. As we have seen, some terminates while some do not.
- normal form*: an expression that cannot be reduced anymore.
 - 49 is in normal form, while 7×7 is not.
 - Some expressions do not have a normal form. E.g. *infinity*.
- A corollary of the *Church–Rosser theorem*: an expression has at most one normal form.
 - If two evaluation sequences both terminate, they reach the same normal form.

Evaluation Order

- Applicative order evaluation: starting with the innermost reducible expression (a redex).
- Normal order evaluation: starting with the outermost redex.
- If an expression has a normal form, normal order evaluation delivers it. Hence the name.
- For now you can imagine that Haskell uses normal order evaluation. A way to implement normal order evaluation is called *lazy evaluation*.

2 Functions

Mathematical Functions

- Mathematically, a function is a mapping between arguments and results.
 - A function $f :: A \rightarrow B$ maps each element in A to a unique element in B .
- In contrast, C “functions” are not mathematical functions:
 - ```
int y = 1; int f (x:int) { return ((y++) * x); }
```
- Functions in Haskell have no such *side-effects*: (unconstrained) assignments, IO, etc.
- Why removing these useful features? We will talk about that later in this course.

### 2.1 Using Functions

#### Curried Functions

- Consider again the function *smaller*:

```
smaller :: Int → Int → Int
smaller x y = if x ≤ y then x else y
```

- We sometimes informally call it a function “taking two arguments”.
- Usage: *smaller* 3 4.
- Strictly speaking, however, *smaller* is a function returning a function. The type should be bracketed as  $Int \rightarrow (Int \rightarrow Int)$ .

#### Precedence and Association

- In a sense, all Haskell functions takes exactly one argument.
  - Such functions are often called *curried*.
- Type:  $a \rightarrow b \rightarrow c = a \rightarrow (b \rightarrow c)$ , not  $(a \rightarrow b) \rightarrow c$ .
- Application:  $f\ x\ y = (f\ x)\ y$ , not  $f\ (x\ y)$ .
  - smaller* 3 4 means  $(smaller\ 3)\ 4$ .
  - square square* 3 means  $(square\ square)\ 3$ , which results in a type error.
- Function application binds tighter than infix operators. E.g. *square* 3 + 4 means  $(square\ 3) + 4$ .

## Why Currying?

- It exposes more chances to reuse a function, since it can be partially applied.

```
twice :: (a → a) → (a → a)
twice f x = f (f x)
quad :: Int → Int
quad = twice square
```

- Try evaluating `quad 3`:

```
quad 3
= twice square 3
= square (square 3)
= ...
```

- Had we defined:

```
twice :: (a → a, a) → a
twice (f, x) = f (f x)
```

we would have to write

```
quad :: Int → Int
quad x = twice (square, x)
```

- There are situations where you'd prefer not to have curried functions. We will talk about conversion between curried and uncurried functions later.

## 2.2 Sectioning

### Sectioning

- Infix operators are curried too. The operator `(+)` may have type `Int → Int → Int`.
- Infix operator can be partially applied too.

```
(x ⊕) y = x ⊕ y
(⊕ y) x = x ⊕ y
```

- `(1 +) :: Int → Int` increments its argument by one.
- `(1.0 /) :: Float → Float` is the “reciprocal” function.
- `(/ 2.0) :: Float → Float` is the “halving” function.

## Infix and Prefix

- To use an infix operator in prefix position, surrounded it in parentheses. For example, `(+) 3 4` is equivalent to `3 + 4`.
- Surround an ordinary function by back-quotes (not quotes!) to put it in infix position. E.g. `3 `mod` 4` is the same as `mod 3 4`.

## Function Composition

- Functions composition:

```
(.) :: (b → c) → (a → b) → (a → c)
(f · g) x = f (g x)
```

- E.g. another way to write `quad`:

```
quad :: Int → Int
quad = square · square
```

- Some important properties:

- `id · f = f = f · id`, where `id x = x`.
- `(f · g) · h = f · (g · h)`.

## 2.3 Definitions

### Guarded Equations

- Recall the definition:

```
smaller :: Int → Int → Int
smaller x y = if x ≤ y then x else y
```

- We can also write:

```
smaller :: Int → Int → Int
smaller x y | x ≤ y = x
 | x > y = y
```

- Equivalently,

```
smaller :: Int → Int → Int
smaller x y | x ≤ y = x
 | otherwise = y
```

- Helpful when there are many choices:

```
signum :: Int → Int
signum x | x > 0 = 1
 | x == 0 = 0
 | x < 0 = -1
```

Otherwise we'd have to write

```
signum x = if x > 0 then 1
 else if x == 0 then 0 else -1
```

## $\lambda$ Expressions

- Since functions are first-class constructs, we can also construct functions in expressions.
- A  $\lambda$  expression denotes an anonymous function.
  - $\lambda x \rightarrow e$ : a function with argument  $x$  and body  $e$ .
  - $\lambda x \rightarrow \lambda y \rightarrow e$  abbreviates to  $\lambda x y \rightarrow e$ .
  - In ASCII, we write  $\lambda$  as `\`
- Yet another way to define *smaller*:

$smaller :: Int \rightarrow Int \rightarrow Int$   
 $smaller = \lambda x y \rightarrow \text{if } x \leq y \text{ then } x \text{ else } y$

- Why  $\lambda$ s? Sometimes we may want to quickly define a function and use it only once.
- In fact,  $\lambda$  is a more primitive concept.

## Local Definitions

There are two ways to define local bindings in Haskell.

- **let**-expression:

$f :: Float \rightarrow Float \rightarrow Float$   
 $f\ x\ y = \text{let } a = (x + y)/2$   
 $\quad\quad b = (x + y)/3$   
 $\quad\quad \text{in } (a + 1) \times (b + 2)$

- **where**-clause:

$f :: Int \rightarrow Int \rightarrow Int$   
 $f\ x\ y \mid x \leq 10 = x + a$   
 $\quad\quad \mid x > 10 = x - a$   
 $\quad\quad \text{where } a = \text{square } (y + 1)$

- **let** can be used in expressions (e.g.  $1 + (\text{let..in..})$ ), while **where** qualifies multiple guarded equations.

## 3 Types

### Types

- The universe of values is partitioned into collections, called *types*.
- Some basic types: *Int*, *Float*, *Bool*, *Char*...
- Type “constructors”: functions, lists, trees ...to be introduced later.

- Operations on values of a certain type might not make sense for other types. For example: *square square 3*.
- Strong typing: the type of a well-formed expression can be deduced from the constituents of the expression.
  - It helps you to detect errors.
  - More importantly, programmers may consider the types for the values being defined before considering the definition themselves, leading to clear and well-structured programs.

## Polymorphic Types

- Suppose  $\text{square} :: Int \rightarrow Int$  and  $\text{sqrt} :: Int \rightarrow Float$ .
  - $\text{square} \cdot \text{square} :: Int \rightarrow Int$
  - $\text{sqrt} \cdot \text{square} :: Int \rightarrow Float$
- The  $(\cdot)$  operator has different types in the two expressions:
  - $(\cdot) :: (Int \rightarrow Int) \rightarrow (Int \rightarrow Int) \rightarrow (Int \rightarrow Int)$
  - $(\cdot) :: (Int \rightarrow Float) \rightarrow (Int \rightarrow Int) \rightarrow (Int \rightarrow Float)$
- To allow  $(\cdot)$  to be used in many situations, we introduce type variables and let its type be:  $(b \rightarrow c) \rightarrow (a \rightarrow b) \rightarrow (a \rightarrow c)$ .

## Summary So Far

- Functions are essential building blocks in a Haskell program. They can be applied, composed, passed as arguments, and returned as results.
- Types sometimes guide you through the design of a program.
- Equational reasoning: let the symbols do the work!

## Recommended Textbooks

- *Introduction to Functional Programming using Haskell* [Bir98]. My recommended book. Covers equational reasoning very well.
- *Programming in Haskell* [Hut07]. A thin but complete textbook.

## Online Haskell Tutorials

- *Learn You a Haskell for Great Good!* [Lip11], a nice tutorial with cute drawings!
- *Yet Another Haskell Tutorial* [DI02].
- *A Gentle Introduction to Haskell* by Paul Hudak, John Peterson, and Joseph H. Fasel: a bit old, but still worth a read. [HPF00]
- *Real World Haskell* [OSG98]. Freely available online. It assumes some basic knowledge of Haskell, however.

## References

- [Bir98] Richard S. Bird. *Introduction to Functional Programming using Haskell*. Prentice Hall, 1998.
- [DI02] Hal Daume III. Yet another haskell tutorial. <http://en.wikibooks.org/wiki/Haskell/YAHT>, 2002.
- [HPF00] Paul Hudak, John Peterson, and Joseph Fasel. A gentle introduction to haskell, version 98. <http://www.haskell.org/tutorial/>, 2000.
- [Hut07] Graham Hutton. *Programming in Haskell*. Cambridge University Press, 2007.
- [Lip11] Miran Lipovača. *Learn You a Haskell for Great Good!* No Starch Press, 2011. Available online at <http://learnyouahaskell.com/>.
- [OSG98] Bryan O’Sullivan, Don Stewart, and John Goerzen. *Real World Haskell*. O’Reilly, 1998. Available online at <http://book.realworldhaskell.org/>.

## A GHCi Commands

|                                                    |                                                                                                                                         |
|----------------------------------------------------|-----------------------------------------------------------------------------------------------------------------------------------------|
| <code>&lt;statement&gt;</code>                     | evaluate/run <code>&lt;statement&gt;</code>                                                                                             |
| <code>:</code>                                     | repeat last command                                                                                                                     |
| <code>:\{ \n ..lines.. \n: \} \n }</code>          | multiline command                                                                                                                       |
| <code>:add [*]&lt;module&gt; ...</code>            | add module(s) to the current target set                                                                                                 |
| <code>:browse[!] [[*]&lt;mod&gt;]</code>           | display the names defined by module <code>&lt;mod&gt;</code> (!: more details; *: all top-level names)                                  |
| <code>:cd &lt;dir&gt;</code>                       | change directory to <code>&lt;dir&gt;</code>                                                                                            |
| <code>:cmd &lt;expr&gt;</code>                     | run the commands returned by <code>&lt;expr&gt;::IO String</code>                                                                       |
| <code>:ctags[!] [&lt;file&gt;]</code>              | create tags file for Vi (default: "tags") (!: use regex instead of line number)                                                         |
| <code>:def &lt;cmd&gt; &lt;expr&gt;</code>         | define command <code>:&lt;cmd&gt;</code> (later defined command has precedence, <code>::&lt;cmd&gt;</code> is always a builtin command) |
| <code>:edit &lt;file&gt;</code>                    | edit file                                                                                                                               |
| <code>:edit</code>                                 | edit last module                                                                                                                        |
| <code>:etags [&lt;file&gt;]</code>                 | create tags file for Emacs (default: "TAGS")                                                                                            |
| <code>:help, :?</code>                             | display this list of commands                                                                                                           |
| <code>:info [&lt;name&gt; ...]</code>              | display information about the given names                                                                                               |
| <code>:issafe [&lt;mod&gt;]</code>                 | display safe haskell information of module <code>&lt;mod&gt;</code>                                                                     |
| <code>:kind &lt;type&gt;</code>                    | show the kind of <code>&lt;type&gt;</code>                                                                                              |
| <code>:load [*]&lt;module&gt; ...</code>           | load module(s) and their dependents                                                                                                     |
| <code>:main [&lt;arguments&gt; ...]</code>         | run the main function with the given arguments                                                                                          |
| <code>:module [+/-] [*]&lt;mod&gt; ...</code>      | set the context for expression evaluation                                                                                               |
| <code>:quit</code>                                 | exit GHCi                                                                                                                               |
| <code>:reload</code>                               | reload the current module set                                                                                                           |
| <code>:run function [&lt;arguments&gt; ...]</code> | run the function with the given arguments                                                                                               |
| <code>:script &lt;filename&gt;</code>              | run the script <code>&lt;filename&gt;</code>                                                                                            |
| <code>:type &lt;expr&gt;</code>                    | show the type of <code>&lt;expr&gt;</code>                                                                                              |
| <code>:undef &lt;cmd&gt;</code>                    | undefine user-defined command <code>:&lt;cmd&gt;</code>                                                                                 |
| <code>!:&lt;command&gt;</code>                     | run the shell command <code>&lt;command&gt;</code>                                                                                      |

### Commands for debugging

|                                                           |                                                                   |
|-----------------------------------------------------------|-------------------------------------------------------------------|
| <code>:abandon</code>                                     | at a breakpoint, abandon current computation                      |
| <code>:back</code>                                        | go back in the history (after <code>:trace</code> )               |
| <code>:break [&lt;mod&gt;] &lt;l&gt; [&lt;col&gt;]</code> | set a breakpoint at the specified location                        |
| <code>:break &lt;name&gt;</code>                          | set a breakpoint on the specified function                        |
| <code>:continue</code>                                    | resume after a breakpoint                                         |
| <code>:delete &lt;number&gt;</code>                       | delete the specified breakpoint                                   |
| <code>:delete *</code>                                    | delete all breakpoints                                            |
| <code>:force &lt;expr&gt;</code>                          | print <code>&lt;expr&gt;</code> , forcing unevaluated parts       |
| <code>:forward</code>                                     | go forward in the history (after <code>:back</code> )             |
| <code>:history [&lt;n&gt;]</code>                         | after <code>:trace</code> , show the execution history            |
| <code>:list</code>                                        | show the source code around current breakpoint                    |
| <code>:list identifier</code>                             | show the source code for <code>&lt;identifier&gt;</code>          |
| <code>:list [&lt;module&gt;] &lt;line&gt;</code>          | show the source code around line number <code>&lt;line&gt;</code> |
| <code>:print [&lt;name&gt; ...]</code>                    | prints a value without forcing its computation                    |
| <code>:sprint [&lt;name&gt; ...]</code>                   | simplified version of <code>:print</code>                         |
| <code>:step</code>                                        | single-step after stopping at a breakpoint                        |

|                                  |                                                                                 |
|----------------------------------|---------------------------------------------------------------------------------|
| <code>:step &lt;expr&gt;</code>  | single-step into <code>&lt;expr&gt;</code>                                      |
| <code>:steplocal</code>          | single-step within the current top-level binding                                |
| <code>:stepmodule</code>         | single-step restricted to the current module                                    |
| <code>:trace</code>              | trace after stopping at a breakpoint                                            |
| <code>:trace &lt;expr&gt;</code> | evaluate <code>&lt;expr&gt;</code> with tracing on (see <code>:history</code> ) |

## Commands for changing settings

|                                                |                                                           |
|------------------------------------------------|-----------------------------------------------------------|
| <code>:set &lt;option&gt; ...</code>           | set options                                               |
| <code>:seti &lt;option&gt; ...</code>          | set options for interactive evaluation only               |
| <code>:set args &lt;arg&gt; ...</code>         | set the arguments returned by <code>System.getArgs</code> |
| <code>:set prog &lt;progrname&gt;</code>       | set the value returned by <code>System.getProgName</code> |
| <code>:set prompt &lt;prompt&gt;</code>        | set the prompt used in GHCi                               |
| <code>:set editor &lt;cmd&gt;</code>           | set the command used for <code>:edit</code>               |
| <code>:set stop [&lt;n&gt;] &lt;cmd&gt;</code> | set the command to run when a breakpoint is hit           |
| <code>:unset &lt;option&gt; ...</code>         | unset options                                             |

## Options for `:set` and `:unset`

|                             |                                                                                                                                                                                                   |
|-----------------------------|---------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------|
| <code>+m</code>             | allow multiline commands                                                                                                                                                                          |
| <code>+r</code>             | revert top-level expressions after each evaluation                                                                                                                                                |
| <code>+s</code>             | print timing/memory stats after each evaluation                                                                                                                                                   |
| <code>+t</code>             | print type after evaluation                                                                                                                                                                       |
| <code>-&lt;flags&gt;</code> | most GHC command line flags can also be set here (eg. <code>-v2</code> , <code>-fglasgow-exts</code> , etc). For GHCi-specific flags, see User's Guide, Flag reference, Interactive-mode options. |

## Commands for displaying information

|                                    |                                                                                                              |
|------------------------------------|--------------------------------------------------------------------------------------------------------------|
| <code>:show bindings</code>        | show the current bindings made at the prompt                                                                 |
| <code>:show breaks</code>          | show the active breakpoints                                                                                  |
| <code>:show context</code>         | show the breakpoint context                                                                                  |
| <code>:show imports</code>         | show the current imports                                                                                     |
| <code>:show modules</code>         | show the currently loaded modules                                                                            |
| <code>:show packages</code>        | show the currently active package flags                                                                      |
| <code>:show language</code>        | show the currently active language flags                                                                     |
| <code>:show &lt;setting&gt;</code> | show value of <code>&lt;setting&gt;</code> , which is one of <code>[args, prog, prompt, editor, stop]</code> |
| <code>:showi language</code>       | show language flags for interactive evaluation                                                               |